



中功率霧化模組

BMN31K522

Arduino Library V1.0.1 說明

版本：V1.02 日期：2023-12-28

www.bestmodulescorp.com

目錄

簡介	3
Arduino Lib 函式	3
Arduino Lib 下載及安裝	4
Arduino 範例	5
範例 : controlAtomization	5

簡介

BMN31K522 是倍創推出的中功率霧化模組，使用 UART 通訊方式。本文檔對 BMN31K522 的 Arduino Lib 函式、Arduino Lib 安裝方式進行說明；範例演示霧化功能和霧化檔位切換功能。

Arduino Lib 函式

Arduino Lib 名稱：BMN31K522		Lib 版本：V1.0.1
構造成函式 & 初始化		
1	BMN31K522(HardwareSerial *theSerial=&Serial)	
	描述	構造成函式，使用 HW Serial 介面
	參數	*theSerial：選擇 HW Serial 介面 (預設 Serial 介面)
	返回值	—
	備註	—
2	BMN31K522(uint8_t rxPin, uint8_t txPin)	
	描述	構造成函式，使用 SW Serial 介面
	參數	rxPin：TX 連接的腳位 txPin：RX 連接的腳位
	返回值	—
	備註	—
3	void begin()	
	描述	模組初始化
	參數	—
	返回值	void
	備註	鮑率固定為 9600
功能函式		
4	uint8_t setAtomizationRank(uint8_t rank)	
	描述	設定霧化檔位
	參數	rank：霧化檔位，範圍 0~255 (PWM 脈寬)
	返回值	執行情況： 0x00：成功 0xFF：失敗
	備註	—
5	uint8_t setAtomizationTime(uint8_t times)	
	描述	設定霧化定時時間
	參數	times：定時時間，範圍 0~255，單位：min
	返回值	執行情況： 0x00：成功 0xFF：失敗
	備註	—


6	uint8_t getAtomizationRank()	
	描述	獲取霧化檔位值
	參數	—
	返回值	霧化檔位 0~255
7	uint8_t getAtomizationTime()	
	描述	獲取霧化定時剩餘時間
	參數	—
	返回值	定時剩餘時間 (單位：min)
8	uint8_t getWaterStatus()	
	描述	獲取水位狀態
	參數	—
	返回值	水位狀態： 0：缺水 1：有水
	備註	—

Arduino Lib 下載及安裝


BMN31K522 Library：可參考下面兩種方法安裝 BMN31K522 的 Arduino Library

方式 1：搜索安裝

搜索安裝：Arduino IDE → 草稿碼 → 匯入程式庫 → 管理程式庫 → 搜索 BMN31K522 → 安裝



搜索安裝流程 1




搜索安裝流程 2

方式 2：加入 .ZIP 程式庫 · 需提前下載 .ZIP 程式庫


下載方法：打開倍創官方網站 (<https://www.bestmodulescorp.com/bmn31k522.html>) 檔目錄下的 Arduino 範例程式 (BMN31K522 Library)。

加入 .ZIP 程式庫：Arduino IDE → 草稿碼 → 匯入程式庫 → 加入 .ZIP 程式庫



Arduino 範例

範例：controlAtomization



實物連接示意圖

範例功能描述：檢水有水的條件下，按鍵作為開關，控制霧化模組定時工作 (霧化檔位、定時時間固定)。LED 亮表示霧化開啟；LED 滅表示霧化停止。

1. 範例打開：Arduino IDE → 檔案 → 範例 → Lib 選擇 (BMN31K522) → 選擇範例 (controlAtomization)

2. 範例說明：

a. 構建對象 & 初始化模組

```

#include <BMN31K522.h>
#define SERIAL_DEBUG      (1)      // 0 - 禁用串列調試輸出；1 - 啟用串列調試輸出
#define ATOMIZER_RANK     (255)
#define ATOMIZER_TIME     (1)
BMN31K522 myAtomizer(2, 3);      // 建立對象
byte atomizerLED = 5;           // 霧化器工作 LED
byte atomizerSwitch = 4;       // 霧化器開關 PIN
unsigned long TimeCount_1s = 0; // 1 秒的時間計數
byte waterStatus = 0;          // 缺水狀態：0 - 缺水；1 - 有水
unsigned long switchOnTime = 0; // 開關時間計數
bool atomizerOnFlag = false;    // 霧化器工作標誌
void setup()
{
    if (SERIAL_DEBUG){
        Serial.begin(9600);
    }    myAtomizer.begin();      // 模組初始化
    pinMode(atomizerLED,OUTPUT);
    pinMode(atomizerSwitch,INPUT);
    digitalWrite(atomizerLED,HIGH);
    TimeCount_1s = millis();      // 開始計時
}
    
```

- b. 間隔 1s 獲取一次檢水資料，並在序列埠監視視窗上顯示檢水狀態；掃描按鈕是否按下，判斷是否開啟 / 關閉霧化工作；當霧化開啟時，間隔 1s 獲取霧化定時剩餘時間，當時間為 0 時，定時結束霧化停止；LED 顯示霧化工作狀態，霧化開啟 – 亮，霧化停止 – 滅。

```

void loop() {
    byte switchStatus;
    uint8_t getTime;
    /***** 間隔 1 秒獲取檢水資料 & 剩餘時間 *****/
    if (millis() - TimeCount_1s >= 1000)
    {
        TimeCount_1s = millis();
        /***** 獲取檢水資料 *****/
        waterStatus=myAtomizer.getWaterStatus();
        if (SERIAL_DEBUG)
        {
            if (waterStatus)
            {
                Serial.println( "Atomizer' s water status is: Have water" );
                // 序列埠監視視窗顯示檢水狀態
            }
            else
            {
                Serial.println( "Atomizer' s water status is: No water" );
                // 序列埠監視視窗顯示檢水狀態
            }
        }
        /***** 獲取剩餘時間 *****/
        if (atomizerOnFlag)
        {
            getTime=myAtomizer.getAtomizationTime();
            if (getTime == 0)
    
```

```
    {
        atomizerOnFlag = false;
        digitalWrite(atomizerLED,HIGH);    // LED 指示霧化工作狀態
    }
    if (SERIAL_DEBUG)
    {
        Serial.print( "Atomizer work remaining time is:" );
        // 序列埠監視視窗顯示霧化定時剩餘時間
        Serial.println(getTime);
    }
}
}
/***** 按鍵掃描 *****/
if (waterStatus == 1)
{
    switchStatus = digitalRead(atomizerSwitch);
    if (!switchStatus)
    {
        if ((millis() - switchOnTime) > 150) // 開關去抖
        {
            switchOnTime = millis();
            if (!atomizerOnFlag)
            {
                atomizerOnFlag = true;
                digitalWrite(atomizerLED,LOW);
                myAtomizer.setAtomizationRank(ATOMIZER_RANK); // 設定霧化檔位
                myAtomizer.setAtomizationTime(ATOMIZER_TIME); // 設定霧化時間
            }
            else
            {
                atomizerOnFlag = false;
                digitalWrite(atomizerLED,HIGH);    // LED 指示霧化工作狀態
                myAtomizer.setAtomizationRank(0); // 關閉霧化工作
                myAtomizer.setAtomizationTime(0);
            }
        }
    }
    else
    {
        switchOnTime = millis();
    }
}
/***** 缺水處理 *****/
else
{
    if (atomizerOnFlag)
    {
        atomizerOnFlag = false;
        digitalWrite(atomizerLED,HIGH);
    }
}
}
```

Copyright© 2023 by BEST MODULES CORP. All Rights Reserved.

本文件出版時倍創已針對所載資訊為合理注意，但不保證資訊準確無誤。文中提到的資訊僅是提供作為參考，且可能被更新取代。倍創不擔保任何明示、默示或法定的，包括但不限於適合商品化、令人滿意的品質、規格、特性、功能與特定用途、不侵害第三人權利等保證責任。倍創就文中提到的資訊及該資訊之應用，不承擔任何法律責任。此外，倍創並不推薦將倍創的產品使用在會因故障或其他原因而可能會對人身安全造成危害的地方。倍創特此聲明，不授權將產品使用於救生、維生或安全關鍵零組件。在救生 / 維生或安全應用中使用倍創產品的風險完全由買方承擔，如因該等使用導致倍創遭受損害、索賠、訴訟或產生費用，買方同意出面進行辯護、賠償並使倍創免受損害。倍創 (及其授權方，如適用) 擁有本文件所提供資訊 (包括但不限於內容、資料、範例、材料、圖形、商標) 的智慧財產權，且該資訊受著作權法和其他智慧財產權法的保護。倍創在此並未明示或暗示授予任何智慧財產權。倍創擁有不事先通知而修改本文件所載資訊的權利。如欲取得最新的資訊，請與我們聯繫。